
sur le lien suivant, Il y a de nombreux programmes sur l’eclairage vélo ….et pas de synthèse….
https://forum.arduino.cc/t/eclairage-a-led-de-velo-3w-tracker-atmega-esp32-maker/567964/68
un blog 
https://iutgenieelectrique.blogspot.com/2021/12/de-led-de-puissance-regulation-pwm.html
un forum
http://velorizontal.1fr1.net/t16874-eclairage-a-del-pour-velo-led-light-for-bike-partie-1
le fichier de l’éclairage arriere clignotant
https://www.fichier-doc.fr/2021/12/05/eclairage-velo-arrirere-rear-bike-arduino-pro-/

Cahier des charges :
Realiser un éclairage arriere autonome avec une led de puissance pour être vue de très loin et de jour et 3 petites leds pour être les côtés de nuit. Le plus léger possible, le plus éclairant possible (2Watt minimum)  avec une autonomie  minimale de 10h et pour un prix faible.

Mais de nombreuses questions subsistent:
Pourquoi faut-il une régulation de courant et un hacheur ?
Pourquoi vaut-t-il mieux utiliser des transistors canal P que le canal N pour commander les leds ?
Comment vérifie-t-on que le transistor fonctionne bien ainsi que la carte electronique ?
Comment est gérer le mode flash ?
Quelle est la consommation et l’autonomie au final ?
Comment vérifie-t-on le mode sleep ?
Comment est gérer l’état de charge de la batterie ?
 Le hacheur
Le hacheur permet de faire varier le courant dans la led en fonction du rapport cyclique.
La régulation de courant permet de limiter le courant dans la led malgré la variation de tension de la batterie qui se décharge. La régulation de courant permet d’avoir un flux lumineux constant et diffèrent mode de fonctionnement.
La fréquence de 64KHz du timer 0 pour minimiser l’inductance

Le transistor canal N a sa source à la masse donc facile pour le commander. Mais difficile de mesurer le courant car la  résistance de mesure qui est à la masse pour être lu par le microcontrôleur est haché. Cette tension image du courant haché doit être filtrée par un gros condensateur.
Le transistor canal P a sa broche « source » relié à l’alimentation, donc la résistance de mesure du courant R20 est à la masse et l’Arduino peur mesurer le courant dans la led
	L’Arduino pro mini est bien mieux car consomme moins que l’Arduino nano et un interrupteur permet de couper la consommation de la batterie. le micro permet de gérer le mode flash et la régulation.
La batterie alimentera directement le processeur qui peut fonctionner de 3.3V à 4.2V. Donc évitant les pertes dans un régulateur de tension.

Fig 1. schéma électrique 



 le schéma électrique :

Choix de la led blanche
Les distributeurs de composants discrets ont du mal de fournir des leds de puissances et les focales associés en PMMA ou parabolique….
De plus, les grands distributeurs européens mettent à part, les leds et les MCPCB le support de refroidissement
De plus, ces focales demandent une bonne fixation de la mécanique
Il faut dire que la technologie des led est très rapide et avec une obsolescence rapide…..

Mais, il est possible de prendre des petites leds faible angle 5° de demi angle et de les mettre en // avec 20 lumen/W pour un prix 0.3€ unitaire qui son tassez disponible.
Etude page 48 et 45 de la VLCS5830 sur ce lien
https://forum.arduino.cc/t/eclairage-a-led-de-velo-3w-tracker-atmega-esp32-maker/567964/48

[image: ]
Si l’on compare par rapport à une led cree rouge XRE de  60 lumen/W jusqu’à 2.2W donc 1A max pour 2.5€
http://velorizontal.1fr1.net/t16874p550-eclairage-a-del-pour-velo-led-light-for-bike-partie-1
https://fr.aliexpress.com/item/32987437612.html?spm=a2g0o.store_pc_groupList.8148356.55.11567ab6nncoSF

Donc si l’on désire 40 lumens, il faudra 20 petites led qui vont couter 6€ donc 2 fois plus chères que la XRE mais il n’y aura pas besoin d’optique.
Ces 20 leds de diamètre 5mm vont prendre un surface 2.5cm*2cm
Il est possible de mettre quelque led avec des demi-angles à 45° pour être plus vu sur le côté aussi
Etant donné que la tension de seuil de la led rouge est de 2.2V à 0.05A, il faut tous les mettre en //.
La puissance absorbée par led sera
P=2.2V*0.1A*20=4.4W,     
Avec une résistance de limitation le courant de 10ohm doit être utilisé avec une batterie lithium de 4,2V à 3V.
La puissance perdue dans la résistance sera P=R.I^2=10*(0.1*20)^2=4W
Donc, utiliser une résistance de limitation est bien idiote, d’où l’utilisation du hacheur.

Pas trouver de led red en 3 mm, ni en CMS avec des faibles angles pour amplifier les lux

  Programme boucle ouverte pour tester electronique  « led_velo_PROmini_sleep_V1 »
Un programme tout simple permet de vérifier le bon fonctionnement des hacheurs série.
Il faut diminuer progressivement le rapport cyclique pour augmenter le courant.
La led D13 qui est sur la carte de l’Arduino pro  sera en mode clignotant permet de vérifier la routine d’interruption.

	#include <SoftwareSerial.h>
#include <TimerOne.h>

#define led13   13       // 13

byte PWM=0;
int temps=0;
int temps1=0;

float Vbatt;
float I=0,erreur=0,kp=1,ki=7.5,Ic=1,Integral=200;
float I3=0,erreur3=0, Ic3=0.1,Integral3=200;

bool envoie;


void callback()  {    //1ms
 temps++;
 temps1++;
 if (temps1>100)  {envoie=1;}   //envoie des données toutes les 0.1secondes
}  //fin callback



void setup() {
pinMode(led13, OUTPUT);   //
pinMode(5, OUTPUT);      //PWM
pinMode(6, OUTPUT); 
Serial.begin(9600);

analogWrite(5,255);
analogWrite(6,255);

 Timer1.initialize(1000);           // initialize timer1, and set a 0,1 second period =>  100 000  pour 0.01s  10 000
 Timer1.attachInterrupt(callback);   // attaches callback() as a timer overflow interrupt

analogReference(INTERNAL);  //1.1V 
//  TCCR2B = (TCCR2B & 0b11111000) | 0x01;      //pin 3 et 11  32khz    http://playground.arduino.cc/Main/TimerPWMCheatsheet  
   TCCR0B = (TCCR0B & 0b11111000) | 0x01;      //pin 5 et 6  64khz    http://playground.arduino.cc/Main/TimerPWMCheatsheet                                       
}


void loop() {  

// avec le transitor canal P, PWM inversé
analogWrite(6,255);    //PWM   led power 3Watt
I=analogRead(A1);
I=I*0.002148;       //1.1/(1024*0.5)    mesure du courant max 2.2A

analogWrite(5,110);    //PWM 3led power 0.1Watt
I3=analogRead(A3);
I3=I3*0.002148; 

if (envoie==1) {
Vbatt=analogRead(A0);       //1.1*4.2/1024
Vbatt=Vbatt*0.00451;
Serial.print("V=");Serial.print(Vbatt,2);Serial.print(";");
Serial.print("I=");Serial.print(I,3);Serial.print(";");
Serial.print("I3=");Serial.print(I3,3);Serial.println(";");
envoie=0;
}

 if (temps<1000 && temps>0)     {digitalWrite(led13,1); }  //1 seconde allumée

 if (temps>1000)     { digitalWrite(led13,0);}  //1 seconde eteinte
                      
  if (temps>2000)     {temps=0;}
} // fin loop   



Le signal bleu de l’oscilloscope est la commande de l’Arduino correspond à la PWM.
Le signal rouge correspond à la sortie du transistor ou le modèle de simulation n’est pas connu dont la tension VGS seuil et la capacité Ciss. A 64KHz, la saturation a du mal à être effectué. La tension -0.6V du signal E2 correspond à celle de la diode de toue libre D2.
Le signal jaune est le courant dans la led, il est pratiquement constant grâce au filtre de l’inductance

La simulation permet de connaitre les signaux mais ne permet pas de vérifier le courant en fonction du rapport cyclique à cause des résistances parasites du montage et de la variation de la tension de seuil des led en fonction de la température.
Donc, la mesure de courant doit se faire par la liaison série.
Une vérification d’un nouveau système doit toujours être effectuée en boucle ouverte avant de le faire  fonctionner en boucle fermée. Car il peut avoir des divergences de la régulation. 
De plus, il faut vérifier que la mesure  de la variable soit correcte.
Dans notre cas, il faut vérifier que le courant ne soit pas trop discontinu, sinon il faudra le filtrer numériquement et analogiquement.

La dimension de l’inductance pour une valeur fixe de 0.1mH va dépendre du diamètre de fil utilisé et de son blindage. Le diamètre du fil va provoquer une résistance parasite et un échauffement de la bobine et des pertes. De plus, il y aura une saturation de l’inductance si l’on dépasse le courant maximal. Donc des essais doivent être réalisés.

	Pour l’inductance 0.1mH/1.5A/0.17Ω/1.6€
https://fr.rs-online.com/web/p/inductances-a-montage-en-surface/7701003

Le courant dans la résistance shunt CH1 et la tension aux bornes de la diode de la roue libre CH2 en réelle sont les suivantes avec une PWM de 85.
La tension CH2 est bien carrée prouvant que le transistor est bien saturé.
Le courant moyen est bien continu
Imoyen=(3.3*(1-85/255)-2)/0.91ohms=218mA
Avec une résistance de mesure de 0.5 Ω, CH1 moyen est bien de 109mV.
L’ondulation du courant dans la led est de seulement de 25mA.
Théoriquement, l’ondulation aurait du être
I=Vbatt/4*L*F=4V/(4*0.1mH*64K)=0.156A  

Rapport cyclique de 0.66=(255-85)/255




	Pour l’inductance 0.1mH/0.3A/1.5Ω/0.6€ et resistance de mesure 0.5 Ω

Mesure de la resistance de l’inductance 1.1 Ω
https://fr.farnell.com/bourns/srr0603-101kl/inductance-de-puissance-blinde/dp/2333881?ost=2333881
8 leds 8°
Imoyen=((4.8*0.85-0.4*0.15)-2.2)/2.73Ω =584mA   avec increment de temperature de 50°C

L’ondulation du courant dans la led est environ de 300mA négligeable par rapport à 584mA.

Remarque : pour les 2 inductances précédentes de même valeurs sans être saturé, mais de dimensions différentes, l’ondulation est différente ce qui n’est pas normal en théorie.
La puissance perdue dans l’inductance n’est pas  négligeable et sa température va être limite
P=RI^2=1.1*0.58A^2=0.37W   donc une résistance thermique de l’inductance 135°C/W.
De plus, la saturation de l’inductance va entrainer une augmentation de l’ondulation de courant comme on peut l’observer sur la figure suivante

	Toujours pour l’inductance 0.1mH/0.3A/1.5Ω et résistance de mesure 0.5 Ω

Pour un courant moyen de 824mA, l’ondulation de courant est passée à 600mA à cause de la saturation de courant qui fait diminuer l’inductance lorsque le courant augmente

Pour ce courant, l’incrément de température de 80°C

Est-ce qu’une inductance non blindée radiale serait préférable sachant qu’elle soit moins chere mais va telle perturbé le montage?
0.1mH/3A  18.4*15*7mm 0.19 Ω est bien trop grande pour etre sur notre montage
https://fr.rs-online.com/web/p/inductances-a-montage-en-surface/7409508
0.1mH/1.35A/0.7  .7  10*8mm 0.19 Ω  cette inductance passe pour notre cahier des charges
https://fr.rs-online.com/web/p/inductances-traversantes/1633340



	pour l’inductance non blindée radiale 0.1mH/0.9A/0.32Ω/0.4€ et résistance de mesure 0.5 Ω
https://fr.rs-online.com/web/p/inductances-traversantes/6755208

mais la mesure de la résistance de l’inductance est de 0.54 Ω
Pour un courant moyen de 928mA, l’ondulation de courant est passée à 200mA. 

Pour ce courant, l’incrément de température de 60°C

P=RI^2=0.5*0.92A^2=0.42W   donc une résistance thermique de 142°C/W.

Pour 2 petites leds demandant 20mA
	inductance 0.22mH/50mA/6.2Ω 

La tension CH2 est bien carrée prouvant que le transistor est bien saturé.
Imoyen=(3.3*0.57-1.65)/7.7ohms=30mA
L’ondulation du courant dans la led est de 80mA. 
Sur la figure suivante, le régime de courant est limite.
Inductance 1mH/0.2A/2.3Ω
https://fr.farnell.com/bourns-jw-miller/5800-102-rc/high-current-choke-1mh-200ma-10/dp/1103711
Inductance 0.47mH/0.15A/5Ω
https://www.we-online.com/katalog/datasheet/744052471.pdf
inductance radiale 0.47mH/0.15A/6Ω
https://fr.farnell.com/murata-power-solutions/11r474c/standard-inductor-470uh-140ma/dp/1710429

Rapport cyclique de 0.57= (255-110)/255 
Résistance shunt de 0.5ohms





Pour les courants faibles ou l’ondulation n’est pas négligeable.
D’ailleurs, le choix de la consigne de courant ne devra pas être trop faible avec une mesure correcte du courant, pour que la régulation fonctionne 

Alors un condensateur peut être rajouté en // de la résistance 0.5 ohms de mesure de 22µF qui a une fréquence de coupure de 10kHz.
Mais,  un filtrage numérique passe bas peut être ajouté pour avoir un courant de mesure plus stable.

D’ailleurs, il faut mesurer le courant pour savoir quelle sera la variation du courant avec l’échantillonnage du microcontrôleur comme on peut l’observer sur les figures suivantes.

 COMMUNICATION pour changer le PWM ou la consigne
Pour faire des tests, le programme suivant permet de changer la consigne avec l’USB est le suivant.
Il est possible de changer la consigne du courant d’éclairage avec 2 Boutons poussoirs.
L’USB est protégé en courant rentrant, donc s’il y a une alimentation externe de 5.1V alors la puissance viendra de l’alimentation extérieure et non de l’USB.
Par conséquent, il est possible de mesurer la consommation de l’éclairage.
[image: ]
Voici le programme de la communication
	/#include <LiquidCrystal.h>
#include <SoftwareSerial.h>
#include <TimerOne.h>

#define led13   13       // 13

byte PWM=0;
int temps=0;
int temps1=0;

float Vbatt;
float I=0,erreur=0,ki=0.5,Ic=0.1,Integral=120;

bool envoie;

void callback()  {    //1ms
 temps++;
}  //fin callback

void setup() {
pinMode (10, INPUT_PULLUP);
pinMode (11, INPUT_PULLUP);
pinMode(led13, OUTPUT);   //
pinMode(6, OUTPUT);      //PWM
//Serial.begin(115200);
Serial.begin(57600);

analogWrite(6,255);

 Timer1.initialize(5000);           // initialize timer1, and set a 0,1 second period =>  100 000  pour 0.01s  10 000
 Timer1.attachInterrupt(callback);   // attaches callback() as a timer overflow interrupt

analogReference(INTERNAL);  //1.1V 
 
TCCR0B = (TCCR0B & 0b11111000) | 0x01;      //pin 5 et 6  64khz    http://playground.arduino.cc/Main/TimerPWMCheatsheet                                       
}




void loop() { 
  if ( digitalRead(13)== 1 ) {digitalWrite(13,0);}  else {digitalWrite(13,1);} 
I=analogRead(A3);
I=I*0.002148;       //1.1V/(1024*0.5ohms)    mesure du courant max 2.2A
erreur=Ic-I;
Integral= Integral+(erreur*ki); 

if ( (digitalRead(10))==0 ) {Ic=Ic+0.2;}  //changement des valeurs avec BP
if ( digitalRead(11)==0 ) {Ic=Ic-0.2;}

  if (Integral>=255) {Integral=254;}
  if (Integral<=0) {Integral=0;}
  

  if (Integral>=255) Integral=255;
  if (Integral<=0) {Integral=0;}
PWM=255-Integral;      // avec le transitor canal P, PWM inversé
analogWrite(6,PWM);   //PWM   led power 3Watt   




Vbatt=analogRead(A0);       //1.1*4.2/1024
Vbatt=Vbatt*0.00451;
//Serial.print(Vbatt,2);Serial.print(";");

Serial.print(Ic,2);Serial.print(", ");
Serial.print(I,3);Serial.print(", ");          //traceur serie
Serial.print(Integral,0);Serial.println(",");

 while(Serial.available()>0) 
  {
    float lecture=Serial.read();  //LECTURE ASCII
    lecture=(lecture-48);         //48ascii=0
    if (lecture>6) {lecture=6; }  //limitation à 0.6A
    Ic=(lecture/10);  //changement de consigne par le PC
   }
   
  
} // fin loop  



Autonomie et mode sleep avec Wake up timer
L’alimentation de l’éclairage avec une ou des batteries lithium MCN 42g de 2.5A.h 3€ dans un boitier 18650 (18mm de diametre, 650mm de long). La tension de l’alimentation de 4.2V à 3V, elle alimentera directement le microcontrôleur sans utiliser de régulateur. Il est facile de rajouter des batteries en parallèle pour augmenter l’autonomie, au détriment de la masse.

En effet, le microcontrôleur peut fonctionner entre 5V et 3V. Mais il faudra utiliser la tension référence interne pour la conversion analogique.

Pour vérifier la consommation du microcontrôleur et le bon fonctionnement du mode sleep voici le programme qui fait fonctionner la régulation pendant 50ms avec une routine d’interruption du timer 1 de 4ms en mode sleep.
Puis met en mode sleep pendant 1s.

	#include <avr/sleep.h>
#include <avr/power.h>
//http://donalmorrissey.blogspot.com/2011/11/sleeping-arduino-part-4-wake-up-via.html
#define led13   13       // 


byte PWM=255;
int temps=0;
int temps1=0;

float Vbatt;
float I=0,erreur=0,kp=1,ki=7.5,Ic=1,Integral=200;
float I3=0,erreur3=0, Ic3=0.1,Integral3=200;

bool envoie;
bool f_timer;    //f


ISR(TIMER1_OVF_vect)   //TCCR1B = 0x01;  4ms  
{
temps++;  
}


void enterSleep(void)
{
  set_sleep_mode(SLEEP_MODE_IDLE);
//  set_sleep_mode(SLEEP_MODE_ADC);
//  set_sleep_mode(SLEEP_MODE_PWR_DOWN);
  sleep_enable();
  power_adc_disable();
  power_spi_disable();
  power_timer0_disable();
  power_timer2_disable();
  power_twi_disable();  

  sleep_mode();     /* Now enter sleep mode. */
  
  sleep_disable(); /* First thing to do is disable sleep. */
  power_all_enable();     // Re-enable the peripherals. 
}



void setup() {
pinMode(led13, OUTPUT);   //
pinMode(5, OUTPUT);      //PWM
pinMode(6, OUTPUT); 
Serial.begin(9600);

analogWrite(5,255);
analogReference(INTERNAL);  //1.1V 

TCCR1A = 0x00;   
TCNT1=0x0000; 
TCCR1B = 0x01;  //routine interruption 4ms wake up consommation 7.6mA sous 5V
TIMSK1=0x01;   // Enable the timer overlow interrupt

}//fin set up

                                    



void loop() {  

  if(temps<12)
  {
    TCCR0B = (TCCR0B & 0b11111000) | 0x01;  //pin 5 et 6  64khz    
    PWM=0;
    analogWrite(5,PWM);
    digitalWrite(led13,1); 
   }

  if(temps==13)
  {
    TCCR0B = (TCCR0B & 0b11111000) | 0x01;  
    PWM=255;
    analogWrite(5,PWM);
    digitalWrite(led13,0); 
TCCR1B = 0x04;    //routine interruption 1s wake up
    }    
      
 if(temps>=14)
  {
TCCR1B = 0x01;    //renitialisation
temps=0;
    } 
          
    enterSleep();       /* Re-enter sleep mode. */
 
} // fin loop  



Voici la consommation de l’Arduino sans son régulateur avec différent mode
	Led  1.5A*0.05=75mA moyen
	
	
	
	

	Pro mini 16Mhz Sans mode sleep
	12mA sous 4.2V
	
	
	

	Pro mini 16Mhz avec mode sleep
	5.1mA sous 4.2V
	3.1mA  3.6V
	2.2mA 3.2V
	1.7mA  3V


Avec ce tableau, on peut observer que le courant dans le processeur n’est pas si négligeable par rapport à la consommation moyenne de la led.

L’autonomie minimale correspondra à l’équation suivante avec un Arduino Pro
Temps(heure)=2.5A.h/(0.004+0.075)A=2.5A.h/(0.079)A=31h

Le rendement du système numérique correspondra à l’équation suivante 
rendement=(0.075*2V)/(0.004+0.075*2V/3.7)*3.7V=0.15W/(0.018+0.15)W=89%

 Régulation du courant led
Pour avoir un éclairage identique malgré la variation de la tension de la batterie, l’asservissement du courant est nécessaire. 
Mais, il faut que la mesure du courant soit correcte et constant sinon il y aura une variation du courant moyen. 
Il existe plusieurs solutions
Pour avoir une mesure de courant constant, un filtre analogique avec un condensateur sur la résistance shunt, un filtre numérique passe bas sur la mesure de courant, des coefficients du correcteur PI faible qui permet une variation de la PWM faible malgré les erreurs de mesures
Le traceur série permet de vérifier le fonctionnement.
La période d’échantillonnage a été fixée à 5ms pour faire un compteur de mode flash futur de 0.05s allumé et 1 seconde éteinte. 
Attention que le programme ne dépasse pas la période d’échantillonnage !
D’où la led 13 utilisé pour connaitre le temps du programme avec un oscilloscope.
Or, la liaison série est relativement longue. D’ailleurs, avec 2 envoies de données avec 115200 bauds, le temps du déroulement du programme est de 1ms.
Pour filtrer le courant
· il faudrait un condensateur de 22µF en parallèle sur la résistance shunt ce qui n’est pas viable dans le peu de place de notre carte.
· un filtre numérique avec une période d’échantillonnage de 5ms permettrait d’avoir une fréquence de coupure de 10Hz assez facilement.
· une évolution lente de la régulation va être testée.

Le coefficient proportionnel est inutile, il vaut mieux initialiser la valeur intégrale pour atteindre déjà une valeur de courant à l’allumage

	exemple pour un courant de consigne de 0.1A,  ki=0.1 pour Ubatterie=3.3V
la variation de la mesure du courant et celle de la PWM peut être observé sur la figure suivante.
La PWM est fixe.




	[image: ]courant mesuré*1000
temps (5ms/echantillon)
PWM





	exemple pour un courant de consigne de 0.2A,  ki=0.5, pour Ubatterie=3.3V
ou l’on peut observer les variations de la mesure du courant et celle de la PWM.

PWM




Il est possible de simuler la régulation mais il y aura des différences entre la simulation et la pratique.
En effet, le rapport entre le courant dans la led en fonction du rapport cyclique dépend des résistances parasites des composants et de la tension de seuil de la led.
2 solutions pour connaitre la dynamique du courant dans la led :
· diminuer la base de temps de l’oscilloscope
· prendre les valeurs du terminal, créer un fichier CSV et tracer la courbe à l’oscilloscope
La simulation n’est pas en temps réel à cause du hachage qui demande beaucoup de calcul au PC et il y a des risques de bugs
	exemple pour un courant de consigne de 0.5A,  ki=1, pour Ubatterie=4V
malgré les variations du courant, la PWM est relativement stable.
La constante de temps de la régulation est de 100ms donc assez rapide par rapport à œil humain pour atteindre le flux lumineux désiré.

	avec un coefficient ki=2 toujours pour un courant de consigne de 0.5A, alors la constante de temps de la régulation passe à 50ms comme on peut l’observé sur la figure suivante

	toujours pour un courant de consigne de 0.5A, on retrouve les dynamiques simulés précédentes en pratique.
pour un coefficient ki=0.5, alors la constante de temps de la régulation passe à 250ms comme on peut l’observer sur l’oscilloscope suivant.

voici le programme de la régulation :  « led_velo_PROmini_regul »
	#include <SoftwareSerial.h>
#include <TimerOne.h>

#define led13   13       // 13

byte PWM=0;
int temps=0;
int temps1=0;

float Vbatt;
float I=0,erreur=0,ki=1,Ic=0.2,Integral=120;
float I3=0,erreur3=0, Ic3=0.1,Integral3=120;


void callback()  {    //1ms
 temps++;
}  //fin callback

void setup() {
pinMode(led13, OUTPUT);   //
pinMode(5, OUTPUT);      //PWM
pinMode(6, OUTPUT); 
Serial.begin(115200);


analogWrite(5,255);
analogWrite(6,255);

 Timer1.initialize(5000);           // initialize timer1, and set a 0,1 second period =>  100 000  pour 0.01s  10 000
 Timer1.attachInterrupt(callback);   // attaches callback() as a timer overflow interrupt

analogReference(INTERNAL);  //1.1V 
//  TCCR2B = (TCCR2B & 0b11111000) | 0x01;      //pin 3 et 11  32khz    http://playground.arduino.cc/Main/TimerPWMCheatsheet  
   TCCR0B = (TCCR0B & 0b11111000) | 0x01;      //pin 5 et 6  64khz    http://playground.arduino.cc/Main/TimerPWMCheatsheet                                       
}


void loop() { 
  if ( digitalRead(13)== 1 ) {digitalWrite(13,0);}  else {digitalWrite(13,1);} 
I=analogRead(A1);
I=I*0.002148;       //1.1/(1024*0.5)    mesure du courant max 2.2A
erreur=Ic-I;
Integral= Integral+(erreur*ki); 

  if (Integral>=255) Integral=255;
  if (Integral<=0) {Integral=0;}
PWM=255-Integral;      // avec le transitor canal P, PWM inversé
analogWrite(6,PWM);   //PWM   led power 3Watt   


analogWrite(5,255);    //PWM 3led power 0.1Watt
I3=analogRead(A3);
I3=I3*0.002148; 


Vbatt=analogRead(A0);       //1.1*4.2/1024
Vbatt=Vbatt*0.00451;
//Serial.print(Vbatt,2);Serial.print(";");
I=I*1000;
Serial.print(I,0);Serial.print(",");          //traceur serie
Serial.print(Integral,0);Serial.println(",");


} // fin loop  



On peut retrouver le temps des dynamiques avec les equations suivantes
la  fonction transfert en boucle ouverte du courant 

ILed = avec  = *(Ic-ILed) de la regulation

En remplaçant le rapport cyclique  dans la première formule on trouve :

R*ILed = ualim**Ic - ualim**ILed - Vseuil

ILed *(R+ualim*) = ualim**Ic -Vseuil

ILed = 

d’où ILed = 

ILed = 
ILed(P) = 
Théorème de la valeur final pour une réponse indicielle de consigne  :
 = 
D’où Ic = ILed la fonction transfert est validée


 Le mode flash
Pour augmenter l’autonomie de l’éclairage, un mode flash est utilisé avec un éclairage de 0.05s toutes les 1 secondes donc un rapport cyclique de 0.05. 50ms correspond à la persistance rétinienne de l’humain.
Pour faire le mode flash, 2 façons
· utiliser la routine d’interruption timer 1 en comptage et faire la régulation de courant sans mode veuille
· utiliser le mode veuille pendant 950ms

voici la solution simple avec la routine d’interruption
	#include <SoftwareSerial.h>
#include <TimerOne.h>

#define led13   13       // 13

byte PWM=0;
int temps=0;

float Vbatt;
float I=0,erreur=0,ki=1,Ic=0.2,Integral=120;
float I3=0,erreur3=0, Ic3=0.1,Integral3=120;

bool envoie;


void callback()  {    //1ms
 temps++;
}  //fin callback

void setup() {

   TCCR0B = (TCCR0B & 0b11111000) | 0x01;    //pin 5 et 6  64khz    
pinMode(led13, OUTPUT);   //
pinMode(5, OUTPUT);      //PWM
pinMode(6, OUTPUT); 
analogWrite(5,255);
analogWrite(6,255);
//Serial.begin(115200);
Serial.begin(57600);
//Serial.begin(19200);


 Timer1.initialize(5000);           // initialize timer1, and set a 0,1 second period =>  100 000  pour 0.01s  10 000
 Timer1.attachInterrupt(callback);   // attaches callback() as a timer overflow interrupt

analogReference(INTERNAL);  //1.1V 
                                     
}


void loop() { 
  if ( digitalRead(13)== 1 ) {digitalWrite(13,0);}  else {digitalWrite(13,1);} 

if (temps<=10)   {   //duree 50ms
I=analogRead(A1);
I=I*0.002148;       //1.1/(1024*0.5)    mesure du courant max 2.2A
erreur=Ic-I;
Integral= Integral+(erreur*ki); 

  if (Integral>=255) Integral=255;
  if (Integral<=0) {Integral=0;}
//Integral=255;  
PWM=255-Integral;      // avec le transitor canal P, PWM inversé
analogWrite(6,PWM);   //PWM   led power 3Watt   

analogWrite(5,255);    //PWM 3led power 0.1Watt
I3=analogRead(A3);
I3=I3*0.002148; 

Vbatt=analogRead(A0);       //1.1*4.2/1024
Vbatt=Vbatt*0.00451;
}

if (temps>11 && temps<190 )  {   //duree 950ms
analogWrite(5,255);   //led etteinte
analogWrite(6,255);
}

if (temps>190 )   { temps=0;   }
  
} // fin loop  




 Etat de la batterie et  Sécurisation 
Un interrupteur est utilisé pour couper l’alimentation et ne pas décharger la batterie à 100% pour ne pas la détruire. 
Lors de la mise sous tension, pour indiquer l’état de la batterie, la led clignotera  X fois/5 avec une période de 0.5s pour indiquer l’état de charge de la batterie, puis un éclairage de 3 secondes en continu sera effectué pour indiquer la fin de la communication. Puis, le mode flash sera effectué.

De plus, si la tension est inférieure à 3.2V alors le micro se mettra en veuille profonde.

	
Vbatt=analogRead(A0);       
Vbatt=Vbatt*0.00451;          //1.1*4.2/1024
pourcentage=83*Vbatt-250 ;  //capacité energetique en %
if (pourcentage>=80) {flash=5;}   //>4V
if (pourcentage<80 && pourcentage>=60 ) {flash=4;}  //>3.73
if (pourcentage<60 && pourcentage>=40 ) {flash=3;}  //>3.5   
if (pourcentage<40 && pourcentage>=20 ) {flash=2;}  //>3.25
if (pourcentage<20 ) {flash=1;}       
Serial.println(pourcentage);

for (int i = 0; i<flash; i++) {
temps=0;  while (temps<100) {digitalWrite(13,1);analogWrite(6,140);Serial.println(temps);}    
temps=0;  while (temps<100) {digitalWrite(13,0);analogWrite(6,255);Serial.println(temps); }                        
                                 }
temps=0;
while (temps<400) {digitalWrite(13,1);analogWrite(6,140); Serial.println(temps);}   

 analogWrite(6,255);                     
 digitalWrite(13,0);            
 




	On peut observer sur la figure suivante pour une tension de 3.6V, il y a 4 impulsions court puis un impulsion longue et les impulsions à 50ms pour se faire voir.

Changement du courant de consigne lors de l’initialisation avec EEPROM
Un interrupteur est utilisé pour couper l’alimentation et permettra grace à l’EPROM de changer de courant de consigne…..

	#include <EEPROM.h>


//EEPROM.put(adress, data);   
//EEPROM.get(adress,data);    
mode = EEPROM.read(0);   //LECTURE DE L'ADRESSE 0 ou le mode est enregistré
mode++;                  //incrementaion de la valeur mode à chaque reset ou coupure d'alimentation batterie
if (mode>2)  {mode=0;}
EEPROM.write(0, mode);
//choix des modes avec l'eeprom dans le set up
  switch (mode) {
  case 0:Ic=0.2;Integral=120;ki=1;     break;  
  case 1:Ic=0.5;Integral=120;ki=1;     break;   
  case 2:Ic=0.6;Integral=120;ki=1;     break; 
 




Option sécurité température LED
Pour ne pas griller la led à une température de jonction de 150°C, mais aussi pour avoir une bonne performance de la led. En effet, le flux lumineux de a led diminue lorsque sa température augmente.
La mesure de  température est effectué avec un LM35 permet de mesurer la température du radiateur de la led et de limiter le courant.
Si la température est supérieure à 70°C, alors le courant de consigne passera à une valeur faible de 0.2A.
	temperatur=analogRead(A2); //  
temperatur=temperatur/9.3; 
 if (temperatur>70) {Ic=0.2;}   //limitation du courant de consigne
Serial.print(Vbatt);Serial.println(temperatur);



La valeur de 70°C sur le dissipateur est choisi pour ne pas avoir une température de jonction supérieur de 100°C avec le modelé thermique suivant :
…….


	La temperature est affichée sur le terminal, ainsi que la tension batterie

Option Bluetooth
Pour changer de mode et indiquer l’etat de la batterie, une communication par bluetooth est possible


Le PCB
Le PCB a  pratiquement les mêmes dimensions qu’une cellule 18650 lithium.

Le programme global sans les options
Le programme utilise 17% de mémoire flash et 11% de la mémoire dynamique
	#include <SoftwareSerial.h>
#include <TimerOne.h>
#include <avr/sleep.h>
#include <avr/power.h>
#include <EEPROM.h>


#define led13   13       // 13

byte PWM=0;
unsigned int temps=0;
byte mode=0;    // etat du bouton poussoir

float Vbatt,pourcentage;
float I=0,erreur=0,ki=1,Ic=0.2,Integral=120;
float I3=0,erreur3=0, Ic3=0.04,Integral3=120;
float temperatur;

byte flash;
bool flagdemarrage=1;


void callback()  {    //5ms
 temps++;
//  if ( digitalRead(10)== 1 ) {digitalWrite(10,0);}  else {digitalWrite(10,1);} 
}  //fin callback

void setup() {
  
pinMode(led13, OUTPUT);   //
pinMode(10, OUTPUT);
pinMode(5, OUTPUT);      //PWM
pinMode(6, OUTPUT); 
analogWrite(5,255);
analogWrite(6,255);
//Serial.begin(115200);
Serial.begin(57600);
//Serial.begin(19200);

TCCR0B = (TCCR0B & 0b11111000) | 0x01;    //pin 5 et 6  64khz  
analogReference(INTERNAL);  //1.1V 

 Timer1.initialize(5000);           // initialize timer1, and set a 0,1 second period =>  100 000  pour 0.01s  10 000
 Timer1.attachInterrupt(callback);   // attaches callback() as a timer overflow interrupt

Vbatt=analogRead(A0);       
Vbatt=Vbatt*0.00451;          //1.1*4.2/1024
pourcentage=83*Vbatt-250 ;  //capacité energetique en %
if (pourcentage>=80) {flash=5;}   //>4V
if (pourcentage<80 && pourcentage>=60 ) {flash=4;}  //>3.73
if (pourcentage<60 && pourcentage>=40 ) {flash=3;}  //>3.5   
if (pourcentage<40 && pourcentage>=20 ) {flash=2;}  //>3.25
if (pourcentage<20 ) {flash=1;}       
Serial.println(pourcentage);

for (int i = 0; i<flash; i++) {
temps=0;  while (temps<100) {digitalWrite(13,1);analogWrite(6,100);Serial.println(temps);}    
temps=0;  while (temps<100) {digitalWrite(13,0);analogWrite(6,255);Serial.println(temps); }                        
                                 }
temps=0;
while (temps<600) {digitalWrite(13,1);analogWrite(6,100); Serial.println(temps);}   

 analogWrite(6,255);                     
 digitalWrite(13,0);  

//EEPROM.put(adress, data);   
//EEPROM.get(adress,data);    
mode = EEPROM.read(0);   //LECTURE DE L'ADRESSE 0 ou le mode est enregistré
mode++;                  //incrementaion de la valeur mode à chaque reset ou coupure d'alimentation batterie
if (mode>2)  {mode=0;}
EEPROM.write(0, mode);
//choix des modes avec l'eeprom dans le set up
  switch (mode) {
  case 0:Ic=0.2;    break;     //changement de courant de consigne
  case 1:Ic=0.5;    break;   
  case 2:Ic=0.6;    break; 

              }//fin switch
}//fin set up


void loop() { 

Vbatt=analogRead(A0);    //test la capacité de la batterie    
Vbatt=Vbatt*0.00451;  
temperatur=analogRead(A2); //  
temperatur=temperatur/9.3; 
 if (temperatur>70) {Ic=0.2;}   //limitation du courant de consigne
//Serial.print(Vbatt,2);Serial.print(";");Serial.println(temperatur);

if (Vbatt>=3.2)   {         

if (temps<=10)   {   //duree 50ms
  digitalWrite(13,1);
I=analogRead(A1);
I=I*0.002148;       //1.1/(1024*0.5)    mesure du courant max 2.2A
erreur=Ic-I;
Integral= Integral+(erreur*ki); 

  if (Integral>=255) Integral=255;
  if (Integral<=0) {Integral=0;}
//Integral=255;        //pour la simulation courant max
//Integral=0;        //pour la simulation courant mini
PWM=255-Integral;      // avec le transitor canal P, PWM inversé
analogWrite(6,PWM);   //PWM   led power 3Watt   


I3=analogRead(A3);
I3=I3*0.002148; 
erreur3=Ic3-I3;
Integral3= Integral3+(erreur3*ki); 
  if (Integral3>=255) Integral3=255;
  if (Integral3<=0) {Integral3=0;}
PWM=255-Integral;      
analogWrite(5,PWM);   //PWM   3leds    
Serial.print(I3,2);Serial.print(";");Serial.println(Integral3);
}

if (temps>11 && temps<190 )  {   //duree 950ms
   digitalWrite(13,0); 
analogWrite(5,255);   //led etteinte
analogWrite(6,255);
}

if (temps>190 )   { temps=0;}
}   //fin if vbatt

if (Vbatt<3.2)  {
   digitalWrite(13,0); 
analogWrite(5,255);   //led etteinte
analogWrite(6,255);
  SLEEP_MODE_PWR_DOWN ;  
  sleep_enable();  
  sleep_mode();          //il faudra une remise à zero pour reveiller
}
  
} // fin loop   




  Choix de led

Les distributeurs de composants discrets ont du mal de fournir des leds de puissances et les focales associés en PMMA ou parabolique….
De plus, les grands distributeurs européens mettent à part, les leds et les MCPCB le support de refroidissement
De plus, ces focales demandent une bonne fixation de la mécanique
Il faut dire que la technologie des led est très rapide et avec une obsolescence rapide…..

Mais, il est possible de prendre des petites leds faible angle 5° de demi angle et de les mettre en // avec 20 lumen/W pour un prix 0.3€ unitaire qui son tassez disponible.
Etude page 48 et 45 de la VLCS5830 sur ce lien
https://forum.arduino.cc/t/eclairage-a-led-de-velo-3w-tracker-atmega-esp32-maker/567964/48
[image: ]
Si l’on compare par rapport à une led cree rouge XRE de  60 lumen/W jusqu’à 2.2W donc 1A max pour 2.5€
http://velorizontal.1fr1.net/t16874p550-eclairage-a-del-pour-velo-led-light-for-bike-partie-1
https://fr.aliexpress.com/item/32987437612.html?spm=a2g0o.store_pc_groupList.8148356.55.11567ab6nncoSF
[image: ]

Donc si l’on désire 40 lumens, il faudra 20 petites led qui vont couter 6€ donc 2 fois plus chères que la XRE mais il n’y aura pas besoin d’optique.
Ces 20 leds de diamètre 5mm vont prendre un surface 2.5cm*2cm
Il est possible de mettre quelque led avec des demi-angles à 45° pour être plus vu sur le côté aussi
Etant donné que la tension de seuil de la led rouge est de 2.2V à 0.05A, il faut tous les mettre en //.
La puissance absorbée par led sera
P=2.2V*0.1A*20=4.4W,     
Avec une résistance de limitation le courant de 10ohm doit être utilisé avec une batterie lithium de 4,2V à 3V.
La puissance perdue dans la résistance sera P=R.I^2=10*(0.1*20)^2=4W
Donc, utiliser une résistance de limitation est bien idiote, d’où l’utilisation du hacheur.

pas trouver de led red en 3 mm, ni en CMS avec des faibles angles


Voici le nouveau typon avec 24 leds 50 mA qui va demander 1.2A pour 2.2V et 2.6W donc 48lumens

[image: ][image: ]

Le schéma électrique
[image: ]

bouton poussoir
https://fr.farnell.com/schurter/1241-1606-11/bouton-poussoir-court-cms-ip67/dp/1217755




 Boitier Mécanique pour les 20 leds arduino mini
La forme du typon va dépendre du choix des leds, de la batterie, de la commande…..
	Un interrupteur à glissière coulissant soudé permet de limité le câblage
https://fr.farnell.com/alps/stsss9131/emission-laterale-commut-sp-3/dp/1123876
2 possibilités soit le mettre en bas, soit de le mettre en façade avec le PCB

	La charge de la batterie pourrait se faire en USB mais cela rallonger la longueur du boitier. 
Mais avec une alimentation spécialisée à 4.2V pour recharger la batterie, c’est plus facile
https://fr.rs-online.com/web/p/connecteurs-d-alimentation-dc/1739847
largeur 14.24, longueur 18.25

Mais une autre solution est de mettre, les 2 ensembles 18.2*10*6mm à coller
https://fr.rs-online.com/web/p/connecteurs-d-alimentation-dc/7347378
[image: ]
L’éclairage arriere sera fixé par des élastiques à la potence de la selle.
Pour  rattraper l’angle de la potence, l’éclairage sera constitué de 2 pièces qui seront tenues par des vis en plastique M3 qui dépasseront et permettra de les maintenir ensemble.
Un plexiglass de 45x35 de 3mm permettra de voir l’action des leds
donc, 2 boitiers devront être réalisés, 
un premier prototypes avait été realisés.
[image: ]

L’IUT utilise « google sketchup » (*.skp) pour faire les dessins à l’imprimante 3D car il a été free pendant très longtemps et assez intuitif mais maintenant il faut une licence qui est facilement crakable pour que les étudiants l’utilise facilement. 
https://getintopc.com/softwares/sketchup-pro-2021-free-download-9641360/
Différentes versions de boitier peuvent être téléchargé ici     
https://drive.google.com/drive/folders/1zKrQm4ZA3xFkqY4pSR8CqOE0L1wpBW2J?usp=sharing
https://drive.google.com/file/d/15SblK52-jEAfqqNl2XVFZhpNVa1KrztU/view?usp=sharing
https://drive.google.com/file/d/1d1KrxjUUARPaXNATbCPMIi1GF9G0lniv/view?usp=sharing

[bookmark: _GoBack]
Amelioration du boitier : voici ce qu’il faudrait faire

   Consommation et Autonomie
La consommation de l’Arduino pro varie en fonction de la tension d’alimentation.
Avec notre programme clignoteur la consommation est la suivante
	Tension batterie (V)
	Courant (A)
	autonomie
	

	4.2
	11.4mA
	
	

	3.5
	8.3 mA
	
	

	3V
	6mA
	
	



la consommation
Le bilan avec 1 batterie de 1.2A.h avec une masse totale de ? et un encombrement de ?

Le bilan avec 1 batterie 18650 de 2.5A.h avec une masse totale de ? et un encombrement de ?

Le bilan avec 1 batterie 21700 de 4A.h avec une masse totale de ? et un encombrement de ?


 Conclusion de l’éclairage arriere de puissance clignotante
Il existe de nombreuses options pour realiser un éclairage à LED.
Les choix pour un compromis puissance, masse, autonomie n’est pas aisé  à effectuer.
Mais pour minimiser la consommation, des choses simples doivent être réalisé.

Un éclairage ESP 32 TTGO avec OLED va être réalisé.




  Eclairage clignotant pour lampe arriere de velo avec ESP32

Pour un mode clignotant ou la led fonctionne 50ms et un arret de 950ms en mode sleep
La consommation du TTGO et du oled sera 3 fois plus important qu’un arduino pro mini.
	Iled  /20
	I consommation TTGO
	

	0
	0.028A=150mA/20
	

	1A/20=0.05
	0.033A
	


avec une batterie de 1.2A.h/0.33=36heures

avec le schema electrique suivant
[image: ]



Voici le programme   (TTGO_eclairage_BFarriere.ino)
	//eclairage arriere clignotant TTGO+OLED+LED CREE+batt
#include <TFT_eSPI.h>
#include <SPI.h>
#include "BluetoothSerial.h"
BluetoothSerial SerialBT;


TFT_eSPI tft = TFT_eSPI();  // Invoke library

   #define S1       0  //fonctionne à 0L     BP gauche
   #define S2      35  // BP droite
   #define ADC_PIN 34
   
RTC_DATA_ATTR bool flag=0,flagT;    //save data sleep
RTC_DATA_ATTR int  wakeup_reason;
RTC_DATA_ATTR float tensionbatt,consigne, courant, erreur;
RTC_DATA_ATTR float timeduration, pourcentage;
RTC_DATA_ATTR unsigned int  PWM, compteur1, compteur;
RTC_DATA_ATTR unsigned int  compteur2, PWMlimit, i;


RTC_DATA_ATTR float entree,entree1,entree2,entree3, entree4,courantFiltre,sortieI,sortieI1,sortieI2, courant1,courant2 ;
RTC_DATA_ATTR float courantled,Integral;

//fe=1000,fc=10;  //frequence de coupure desiréé  
RTC_DATA_ATTR const float a1=-1.911197, a2=0.914976, Gain=1059;
 

const int ledPin32 = 32;
const int ledChannel = 1;


   void espDelay(int ms)  //****************
{
    esp_sleep_enable_timer_wakeup(ms * 1000);
    esp_sleep_pd_config(ESP_PD_DOMAIN_RTC_PERIPH, ESP_PD_OPTION_ON);  //
  //   esp_sleep_pd_config(ESP_PD_DOMAIN_RTC_PERIPH, ESP_PD_OPTION_AUTO,ESP_PD_DOMAIN_CPU);  //coupe le oled
    esp_light_sleep_start();      // no sleep oled 
}


void setup() {  
pinMode(ADC_PIN, INPUT);
pinMode(2, INPUT);
pinMode(S2, INPUT_PULLUP);
pinMode(14, OUTPUT);
pinMode(26, OUTPUT);

digitalWrite(14,1);
  
    tft.init();
    tft.setRotation(0); //format portrait a l'endroit
    Serial.begin(115200);
//    SerialBT.begin("ESP32test"); //Bluetooth device name  
  
  tft.fillScreen(TFT_BLACK);
  tft.setTextColor(TFT_WHITE,TFT_BLACK); 
  analogReadResolution(12);   //12 bits 
 
ledcSetup(ledChannel, 64000, 10);     //PWM 64kHz  10bits
ledcAttachPin(ledPin32, ledChannel);  //broche 32 PWM
 
}


void loop() {
  
if ( digitalRead(26)== 1 ) {digitalWrite(26,LOW);}  else {digitalWrite(26,HIGH);}  //temps de la boucle avec affichage 7.5ms

//comande de la LED en boucle fermée      100us
       if (digitalRead(S1)==0)  {consigne=consigne+0.1;Integral=500;espDelay(200); }
       if (!(digitalRead(S2)))  {compteur++;consigne=consigne-0.1;
                                 if(consigne<=0){consigne=0; }
                                 espDelay(200);}      
       if (digitalRead(S2)==1)  {compteur=0;}   //extinction afficheur
             
       
/* verification des temps de reponse de regulation du courant avec 100us de bouclage du programme */
/*compteur1++;  
if (compteur1<100) {consigne=0; }   //si consigne=0 alors PWM=0 
if (compteur1>100 && compteur1<700) {consigne=0.4; } 
if( compteur1>700) {compteur1=0; }  
*/
       
 //si appuie du S2 plus de 15 secondes (15 fois la loop de 200ms )alors mise en veuille pour recharge batterie           
     if (compteur>=15){
        esp_sleep_disable_wakeup_source(ESP_SLEEP_WAKEUP_TIMER);    //reveille par le RESET                 
        esp_deep_sleep_start();     //sleep le oled                                  
                      }

  tensionbatt=analogRead(ADC_PIN);          //34 interne, pont diviseur par 2 (100Kohm)
  tensionbatt=(tensionbatt*2*3.63)/(4095);
                          
//mise en veuille si batterie < à 3.2                             
   if (tensionbatt<=3.2) {
    esp_sleep_enable_ext0_wakeup(GPIO_NUM_0,0); 
    esp_deep_sleep_start();     //sleep le oled
    }                         

   for (i=0; i <= 500; i++) {     //duree de la boucle 100us*500=50ms            
entree=analogRead(2);        //resistance shunt de 0.5ohm
entree1=analogRead(2);               
entree2=analogRead(2);
entree3=analogRead(2);
entree4=analogRead(2);

courant2= courant1;
courant1= courant;
sortieI2=sortieI1;      //sortie(n-2)
sortieI1=sortieI;       //sortie(n-1)
courant=(entree4+entree3+entree2+entree1+entree)/4;   //filtre moyenneur
courant=(entree*3.65)/(4096);   //mesure un courant max de 14.6A  

sortieI=(courant+courant1*2+courant2-sortieI1*a1-sortieI2*a2) ;      //filtre passe pas recursif ordre 2
courantFiltre=(sortieI/Gain);


erreur=consigne-courantFiltre;
//if (erreur>0)  {Integral= Integral+1; } 
//if (erreur<=0)  {Integral= Integral-1; }
Integral= Integral+erreur*5;       //Integral= Integral+erreur*ki;
PWMlimit=((1.8+2*consigne)*1024)/3.2;    //limitation de la PWM si mesure de courant inefficace
      if (Integral>=PWMlimit) Integral=PWMlimit;   //exemple pour 0.2A   PWMlimit=640
      if (Integral<=0) {Integral=0;}
      if (Integral>=1024) {Integral=1024;}
      PWM=Integral;
      
if (consigne<=0.09)  {consigne=0;PWM=0;}        

  ledcWrite(ledChannel,PWM);                            
  }  //end for



  pourcentage= map(tensionbatt*10, 30, 42, 0, 100);
  timeduration=(pourcentage*1.2)/((0.02+consigne*0.05)*100);    //0.05*consigne correpond au rapport cyclique 50ms/(950ms+50ms)
  tft.setCursor(0, 0, 4);// (x,y,taille)  //durée 4ms
  tft.print(pourcentage);tft.println("=%"); 
  tft.print(timeduration,1);tft.println("=hour");
  tft.print(consigne,2);tft.println("=Cons");
  tft.print(courantFiltre,3);tft.println("=I   ");
  tft.print(PWM);tft.println("=PWM    "); 


//Serial.print(courantled,3);Serial.print(";");Serial.print(consigne,1);Serial.print(";");Serial.println(PWM); 

//mode clignotant
  ledcWrite(ledChannel,0);  //mettre à 0 la PWM                                 
  espDelay(950);     //ne fonctionne pas si la PWM n'est pas nulle
   
 }//end loop



aller voir le blog pour plus de précision
https://iutgenieelectrique.blogspot.com/2021/12/de-led-de-puissance-regulation-pwm.html

Boitier Mécanique pour les 20 leds TTGO




bibliographie

https://forum.arduino.cc/t/petits-exemples-lcd-horloge-communication-motor-capteur-iut-soissons/647014/163

